

# Getting Started with Maple

(September 2003)

Note: This handout is actually a "Maple Session"; everything in here was created in Maple. The actual Maple commands that I entered are shown in the lines preceded by the ">"; these lines are immediately followed by the resulting output.

[ Simple calculations:

[ > **2+2;**

4

[ Note the semicolon at the end of the line. Each Maple command must end with a semicolon (or a colon).

[ > **(12+3\*5)/4;**

$\frac{27}{4}$

[ Note that Maple does not convert the rational number to its decimal representation. You can find the decimal representation with the "evalf" command.

[ > **evalf(27/4);**

6.750000000

[ Maple has many predefined functions. For example, sine and cosine:

[ > **sin(1.23);**

.9424888019

[ > **cos(0);**

1

[ > **cos(1);**

cos(1)

[ Note that Maple didn't display the numerical value of cos(1). To see the numerical value, use "evalf":

[ > **evalf(cos(1));**

.5403023059

[ To take a square root, you can use the "sqrt" function:

[ > **sqrt(4);**

2

[ Maple has several predefined constants. The one that we will need most is pi. The Maple name of pi that can be used in an input line is "Pi". Maple is *case sensitive*, so the P must be a capital letter. In the output, Maple will show the Greek letter.

[ > **Pi;**

$\pi$

[ > **evalf(Pi);**

3.141592654

```
[ > sin(Pi);
```

```
0
```

Another predefined constant that you will see occasionally is "I". In Maple, "I" is the square root of -1 (a complex number). For example:

```
[ > sqrt(-1);
```

```
I
```

```
[ > sqrt(-16);
```

```
4 I
```

```
[ > I*I;
```

```
-1
```

The function name "exp" is used for the exponential function:

```
[ > exp(0);
```

```
1
```

```
[ > exp(1);
```

```
e
```

Note that the number "e" is displayed in the output as a boldface e. However, you should not use  $e^x$  for the exponential function; use  $\exp(x)$ .

```
[ > evalf(exp(1));
```

```
2.718281828
```

Typically we want to do a lot more than calculate a few numbers. We want to define functions, plot graphs, solve equations, etc. We need to be able to define and store expressions in Maple. This is done with the assignment statement ":=". For example:

```
[ > m := 355/113;
```

```

$$m := \frac{355}{113}$$

```

Now the variable "m" has been created, and it has the value 355/113. We can now use "m" in subsequent commands.

```
[ > evalf(m);
```

```
3.141592920
```

```
[ > sin(m);
```

```

$$\sin\left(\frac{355}{113}\right)$$

```

The value of a variable can be an expression that is more complicated than just a number. For example, suppose we want to work with the expression  $4x^2-1$ . We can create a variable that will hold this expression. I will call the variable "q":

```
[ > q := 4*x^2-1;
```


```

$$q := 4x^2 - 1$$

```

Now there are many things we can do with this expressions. For example, we can plot it, using the "plot" command.

```
[ > plot(q,x=-1..1);
```



We'll take a closer look at the plot command later.


We can also take the derivative with respect to  $x$ , using the "diff" command. I will create a new variable called "dq" that holds the derivative:

```
> dq := diff(q,x);
```

$$dq := 8x$$

We could have done the same things without defining  $q$ . We can give the expression itself as an argument to the functions:

```
> plot(4*x^2-1,x=-1..1);
```



```
> diff(4*x^2-1,x);
```

$$8x$$

[ We can also attempt to solve the equation  $4x^2-1=0$  with the "solve" command:

```
> solve(4*x^2-1=0,x);
```

$$\frac{1}{2}, \frac{-1}{2}$$


[ In this case, the solve command was able to find the solutions.

[ Let's take a closer look at the commands "plot", "diff", and "solve".

## The Plot Command

[ The "plot" command takes several arguments, and has many options. The basic command takes two arguments. The first is the expression (i.e. function) to be plotted, and the second is the range of the independent variable of the expression. For example, the following command plots  $x^2$  on the interval  $-1 < x < 1$ :


```
> plot(x^2,x=-1..1);
```



[ Note that a..b is the standard method for describing a range in Maple.


[ We can also give a third argument to the plot command that specifies the vertical range of the plot. For example:

```
> plot(x^2,x=-1..1,-1..1.5);
```



You can specify more optional arguments. In the next example, I've changed the color to black, used a dashed line, and I've given the plot a title:

```
> plot(x^2,x=-1..1,-1..1.5,title=`y=x^2`,color=black,linestyle=DASH)  
;
```



[ To see more examples, look up "plot" in the Maple Help, or enter the command "?plot".

## The Diff Command

You can take derivatives in Maple with the "diff" command ("diff" stands for "differentiate"). The basic command takes two arguments. The first is the expression to be differentiated, and the second is the variable with respect to which the derivative is to be taken. Maple knows the derivatives of most of the elementary functions.

Here are some examples. The first finds the derivative of  $\sin(t)+4t$ , and the second finds the derivative of  $1/x + \exp(x^2)$ .

[ > `diff(sin(t)+4*t,t);`

$$\cos(t) + 4$$

[ > `diff(1/x + exp(x^2),x);`

$$-\frac{1}{x^2} + 2x e^{(x^2)}$$

[ To take a second or higher order derivative, just list the variable again. Here are some second derivatives:

[ > `diff(sin(t)+4*t,t,t);`

$$-\sin(t)$$

[ > `diff(1/x + exp(x^2),x,x);`

$$2\frac{1}{x^3} + 2e^{(x^2)} + 4x^2 e^{(x^2)}$$

## The Solve Command

The "solve" command can solve many algebraic equations. We saw an example earlier where we solved a quadratic. In the simplest form of the command, the first argument is the equation to be solved, and the second is the variable to be solved for. Here we find the solutions to  $x^2+5x+6=0$ :

[ > `solve(x^2+5*x+6=0,x);`

$$-2, -3$$

[ If there expression does not contain an equals sign, Maple assumes that the expression should be set equal to zero.

[ > `solve(x^2+5*x+6,x);`

$$-2, -3$$

[ Next we solve a cubic polynomial:

[ > `solve(x^3-2*x+1=0,x);`

$$1, -\frac{1}{2} + \frac{1}{2}\sqrt{5}, -\frac{1}{2} - \frac{1}{2}\sqrt{5}$$

[ Let's make one small change to the cubic polynomial:

[ > `solve(x^3-x+1=0,x);`

$$\begin{aligned}
& -\frac{1}{6}(108 + 12\sqrt{69})^{(1/3)} - \frac{2}{(108 + 12\sqrt{69})^{(1/3)}}, \frac{1}{12}(108 + 12\sqrt{69})^{(1/3)} \\
& + \frac{1}{(108 + 12\sqrt{69})^{(1/3)}} + \frac{1}{2}I\sqrt{3} \left( -\frac{1}{6}(108 + 12\sqrt{69})^{(1/3)} + \frac{2}{(108 + 12\sqrt{69})^{(1/3)}} \right), \\
& \frac{1}{12}(108 + 12\sqrt{69})^{(1/3)} + \frac{1}{(108 + 12\sqrt{69})^{(1/3)}} \\
& - \frac{1}{2}I\sqrt{3} \left( -\frac{1}{6}(108 + 12\sqrt{69})^{(1/3)} + \frac{2}{(108 + 12\sqrt{69})^{(1/3)}} \right)
\end{aligned}$$

What a mess! If you look carefully at the above output, you will see that Maple has found three solutions, two of which are complex numbers. (Recall that the symbol I in Maple is the complex number sqrt(-1).)

What happens if the equation has no solution, or if Maple can not find it? Here are some examples.

First, a trivial example with no solution:

```
> solve(1=0,x);
```

Maple gave no output; this means that either there is no solution, or Maple can not find any solutions.

Note that Maple can solve an equation like  $x^2=-1$ , because Maple allows complex numbers:

```
> solve(x^2=-1,x);
```

$I, -I$

Sometimes Maple gives a partial solution:

```
> solve(x*(sin(x)+x+1/4)=0,x);
```

$0, \text{RootOf}(4\_Z + 4 \sin(\_Z) + 1)$


The "RootOf" function means the root of the expression; in the expression,  $\_Z$  is the variable. In this example, Maple found one solution to be 0, but the other "solution" is just the solution to  $4*\_Z + 4*\sin(*\_Z) + 1=0$ . Maple could not find an analytical solution to this equation.

Let's create a function, plot it, and then use "diff" and "solve" to find the critical points.

```
> w := (x^4+3*x^2)/(1+x^4);
```

$$w := \frac{x^4 + 3x^2}{1 + x^4}$$


```
> plot(w,x=-3..3);
```



```
> dw := diff(w,x);
```

$$dw := \frac{4x^3 + 6x}{1+x^4} - \frac{4(x^4 + 3x^2)x^3}{(1+x^4)^2}$$

```
> plot(dw,x=-3..3);
```



```
> solve(dw=0,x);
```

$$0, -\frac{1}{3}\sqrt{3+3\sqrt{10}}, \frac{1}{3}\sqrt{3+3\sqrt{10}}, -\frac{1}{3}\sqrt{3-3\sqrt{10}}, \frac{1}{3}\sqrt{3-3\sqrt{10}}$$

```
>
```

[ Note that two of the five solutions found by "solve" are actually complex numbers.